```
> restart:
  with(plots):
Warning, the name changecoords has been redefined
```

Werkzeuge

Diese Prozedur bestimmt eine Gerade der Sekantenschar an *f* "bei *a*".

```
> s:=proc(f,a)
    verbindungsgerade( [a-1/2,f(a-1/2)], [a+1/2,f(a+1/2)] );
  end proc:
```

Ich hatte mit ein paar Varianten gespielt, diese liefert die Tangente an *f* in *a*.
Damit müssen wir als Hüllkurve immer die ursprüngliche zurückbekommen, was für einen Test gar ist.

```
s2:=proc(f,a,b)
  verbindungsgerade( [a,f(a)], [b,f(b)] );
end proc:
s:=proc(f,a)
  s2(f,a-_h,a+_h)(_x);
  limit( %, _h=0 );
  unapply(%,_x);
end proc:
>
```

Erster Versuch mit der Normalparabel

Die Funktion und das ausgewählte Fenster.

```
> f:=x->x^2:
  window:=[-2..2,-1..5]:
```

Erstmal anschauen.

```
> F:=plot( f, window[1], color=blue, view=window,
  title=sprintf("f(x)=%a",f(x)) ):
  F;
```



Wir zeichnen einige Sekanten.

```
> n:=80:
  S:=display( [seq(
    plot( s( f, (1-i/n)*lhs(window[1])+(i/n)*rhs(window[1]) ),
  window[1], color=yellow ),
    i=0..n )], view=window ):
```

Anschauen zusammen mit der Funktion.

```
> display([F,S], view=window);
```



Man erkennt deutlich, dass die Sekanten irgendwie eine Kurve beschreiben: die sog. Hüllkurve
Unsere Geraden scheinen Tangenten an die Hüllkurve zu sein, und das stimmt auch.

Bestimmen wir die Hüllkurve: dazu schneiden wir die Gerade $s(f, a)$ und $s(f, a+\delta)$ und lassen δ gegen 0 gehen. Zunächst ist das nur ein heuristischer Trick, aber man kann zeigen, dass ziemlich allgemein die Geraden dann gerade Tangenten an die so erhaltene Kurve sind.

```
> hparam:=map( limit, Schnittpunkt( s(f,a), s(f,a+delta) ),
  delta=0 );
```

$$hparam := \left[a, a^2 + \frac{1}{4}\right]$$

Zeichnen wir die Hüllkurve.

```
> H:=saveparamplot( [op(hparam),a=window[1]], thickness=2 ):
```

Und jetzt alles zusammen anschauen.

```
> display( [F,H,S], view=window );
```



Die parametrisierte Form in eine direkte umzuwandeln ist hier trivial, wir verschieben das daher auf später, wenn wir ein Beispiel haben, wo nicht mehr $x = a$ herauskommt.

Zusammen

Jetzt fassen wir all das in einer Prozedur zusammen, damit wir verschiedene Funtionen untersuchen können. Wir richten es so ein, dass die parametrisierte Kurve als Ergebnis zurückgegeben wird, damit wir diese eventuell noch bearbeiten können. Das gewünschte Bild ge wir einfach zwischendurch mit `print` aus.

```
> untersuche:=proc( f::procedure, window::[range,range]
  )::procedure;
    local a,F,n,S,hparam,H;
    F:=plot( f, window[1], color=black,
  title=sprintf("x->%a",f(x)) );
    n:=80:
    S:=display( [seq(
      plot( s( f, (1-i/n)*lhs(window[1])+(i/n)*rhs(window[1])
  ), window[1], color=yellow ),
      i=0..n )] ):
```

```
    hparam:=map( limit, Schnittpunkt( s(f,a), s(f,a+delta) ), 
  delta=0 );
    H:=saveparamplot( [op(hparam),a=window[1]], 
  color=[red,green,blue], thickness=2, args[3..-1] );
    print(display( [F,H,S], view=window ));
    unapply( hparam, a );
  end proc:
```

ACHTUNG: Würden wir `hparam` zurückgeben, so enthielte dies die lokale Variable `a`. Das ist zwar nicht unmittelbar schlimm, aber wir hätten später Schwierigkeiten, für `a` etwas einzusetzen, weil wir immer eine globale Version von `a` ansprechen würden, die aber in `hparam` gar nicht drin steckt.

Viele Beispiele

Zuerst nochmal unser Eingangsbeispiel, um zu sehen, dass alles klappt.

```
> untersuche( x->x^2, [-2..2,-1..5] );
```



$$a \rightarrow \left[a, a^2 + \frac{1}{4}\right]$$

```
> untersuche( x->x^2, [-1..1,-0.5..1] );
```



$$a \rightarrow \left[a, a^2 + \frac{1}{4}\right]$$

```
> H:=untersuche( x->cos(Pi/2*x), [-1..1,-2..2] );
```



$$H := a \to \left[\frac{1}{2} \frac{4 \sin\left(\frac{1}{2}\pi a\right) + 2\, a\, \pi \cos\left(\frac{1}{2}\pi a\right) - \sin\left(\frac{1}{2}\pi a\right)\pi}{\pi \cos\left(\frac{1}{2}\pi a\right)}, \right.$$
$$\left. \frac{1}{2} \frac{-4\sqrt{2}\, \sin\left(\frac{1}{2}\pi a\right)^2 + \sqrt{2}\, \pi \cos\left(\frac{1}{2}\pi a\right)^2 + \sqrt{2}\, \pi \sin\left(\frac{1}{2}\pi a\right)^2}{\pi \cos\left(\frac{1}{2}\pi a\right)} \right]$$

Diese Hüllkurve sieht schon nicht mehr so einfach aus. Hier bedeutet es richtig Arbeit, sie in die Form $x \to \mathrm{h}(x)$ zu bringen.

```
> H(a):
  [solve({x=%[1], y=%[2]}, {a,y})];
```

$$[\,]$$

Maple liefert hier keine Lösung. Bei genauerem Hinsehen kann man die x-Gleichung so vereinfachen:

```
> x=collect( convert(expand(H(a)[1]),tan), tan);
```

$$x = \left(\frac{2}{\pi} - \frac{1}{2}\right)\tan\left(\frac{\pi\, a}{2}\right) + a$$

Diese Gleichung ist transzendent und es ist nicht einmal möglich a als Funktion von x wesentlich anders als durch diese Gleichung zu beschreiben; soll heißen: es gibt keinen Ausdruck, der nur addiert, multipliziert, dividiert, trigonometrische Funktionen, exp und ln verwendet und $a = \mathrm{a}(x)$ liefert.

Es könnte natürlich trotzdem noch sein, dass y eine einfach beschreibbare Funktion von x ist, wenn etwa $y = x$ wäre... Aber auch solche Hoffnung wird hier wohl enttäuscht werden.

Ergebnislose Versuche ...

```
> untersuche( x->cos(Pi/2*x), [-4..4,-2..2] );
```



$$a \to \left[\frac{1}{2}\,\frac{4\sin\left(\frac{1}{2}\pi\, a\right) + 2\, a\,\pi\cos\left(\frac{1}{2}\pi\, a\right) - \sin\left(\frac{1}{2}\pi\, a\right)\pi}{\pi\cos\left(\frac{1}{2}\pi\, a\right)},\right.$$

$$\left.\frac{1}{2}\,\frac{-4\sqrt{2}\,\sin\left(\frac{1}{2}\pi\, a\right)^2 + \sqrt{2}\,\pi\cos\left(\frac{1}{2}\pi\, a\right)^2 + \sqrt{2}\,\pi\sin\left(\frac{1}{2}\pi\, a\right)^2}{\pi\cos\left(\frac{1}{2}\pi\, a\right)}\right]$$

```
> H:=untersuche( x->(x-1)*x*(x+1), [-2..2,-5..5] );
```



$$H := a \to \left[\frac{1}{12}\,\frac{12\,a^2-1}{a},\ \frac{1}{16}\,\frac{16\,a^4+1-8\,a^2}{a}\right]$$

Versuchen wir es hier einmal mit einer expliziten Darstellung.

```
> _EnvExplicit:=true:
  H(a): simplify( [solve( { x=%[1], y=%[2] }, {y,a} )] ):
  S:=map( (z->unapply(z,x)) @ subs, %, y );
```

$$S := \left[x \to -\frac{1}{2}\,x + \frac{1}{18}\sqrt{9\,x^2+3} + \frac{1}{2}\,x^3 + \frac{1}{6}\,x^2\sqrt{9\,x^2+3},\right.$$
$$\left. x \to -\frac{1}{2}\,x - \frac{1}{18}\sqrt{9\,x^2+3} + \frac{1}{2}\,x^3 - \frac{1}{6}\,x^2\sqrt{9\,x^2+3}\right]$$

Na fein, gleich zwei Lösungen. Und das Bild zeigt auch zwei. Könnte also passen...

```
> untersuche( x->(x-1)*x^2*(x+1), [-2..2,-1/2..1] );
```



$$a \to \left[\frac{3\,a\,(4\,a^2-1)}{12\,a^2-1}, \frac{3}{16}\,\frac{64\,a^6-16\,a^4-4\,a^2+1}{12\,a^2-1} \right]$$

```
> H:=untersuche( exp, [-1..1,0..3] );
```



$$H := a \rightarrow \left[\frac{-3\,\mathbf{e}^{(-1/2)} + \mathbf{e}^{(1/2)} + 2\,\mathbf{e}^{(1/2)}\,a - 2\,\mathbf{e}^{(-1/2)}\,a}{2\,\mathbf{e}^{(1/2)} - 2\,\mathbf{e}^{(-1/2)}}, \frac{\mathbf{e}^{a}\,\mathbf{e} - 2\,\mathbf{e}^{a} + \mathbf{e}^{a}\,\mathbf{e}^{(-1)}}{\mathbf{e}^{(1/2)} - \mathbf{e}^{(-1/2)}} \right]$$

Versuchen wir auch hier eine explizite Darstellung ...

```
> untersuche( x->exp(x^2), [-1..1,0..3] );
```



$$a \rightarrow \left[\frac{-1 + (\mathbf{e}^a)^2 + 4\,(\mathbf{e}^a)^2\,a^2 - 4\,a^2}{2\,(\mathbf{e}^a)^2 + 4\,(\mathbf{e}^a)^2\,a + 2 - 4\,a}, \frac{\mathbf{e}^{(a^2)}\,\mathbf{e}^{(1/4)} + \mathbf{e}^{(a^2)}\,(\mathbf{e}^a)^4\,\mathbf{e}^{(1/4)}}{(\mathbf{e}^a)^3 + 2\,(\mathbf{e}^a)^3\,a + \mathbf{e}^a - 2\,a\,\mathbf{e}^a} \right]$$

```
> untersuche( x->exp(-x^2), [-2..2,-0.5..1.5] );
```



$$a \rightarrow \left[\frac{-3\,(\mathbf{e}^{a})^{2}+3-4\,a^{2}+4\,(\mathbf{e}^{a})^{2}\,a^{2}}{-2-4\,a-2\,(\mathbf{e}^{a})^{2}+4\,(\mathbf{e}^{a})^{2}\,a}, \frac{-4\,(\mathbf{e}^{a})^{2}\,\mathbf{e}^{(-1/4)}+(\mathbf{e}^{a})^{4}\,\mathbf{e}^{(-1/4)}+\mathbf{e}^{(-1/4)}}{-\mathbf{e}^{(a^{2})}\,\mathbf{e}^{a}-2\,\mathbf{e}^{(a^{2})}\,\mathbf{e}^{a}\,a-\mathbf{e}^{(a^{2})}\,(\mathbf{e}^{a})^{3}+2\,\mathbf{e}^{(a^{2})}\,(\mathbf{e}^{a})^{3}\,a} \right]$$

```
> H:=untersuche( log, [0.6..2,-1..1] );
```



$$H := a \rightarrow \left[\ln(2\,a-1)\,a^{2}-\frac{1}{4}\ln(2\,a-1)-\ln(2\,a+1)\,a^{2}+\frac{1}{4}\ln(2\,a+1)+2\,a,\right.$$

$$-\ln(2\,a-1)^{2}\,a^{2}+\frac{1}{4}\ln(2\,a-1)^{2}+2\ln(2\,a-1)\ln(2\,a+1)\,a^{2}-\frac{1}{2}\ln(2\,a-1)\ln(2\,a+1)$$

$$-\ln(2)-\ln(2\,a-1)\,a+\frac{1}{2}\ln(2\,a-1)+\ln(2\,a+1)\,a+\frac{1}{2}\ln(2\,a+1)-\ln(2\,a+1)^{2}\,a^{2}$$

$$\left.+\frac{1}{4}\ln(2\,a+1)^{2}\right]$$

Na? Wer traut sich?

```
> #H(a): S:=solve( {x=%[1],y=%[2]}, {y,a} ): Y:=(subs( %, y ));
```

Hoppla, viel Spaß beim Lesen ...

```
> dsolve( { D(g)(x)=5*g(x)*(1-g(x)), g(0)=1/2 } );
  untersuche( unapply(subs(%,g(x)),x), [-1..1,0..1] );
```

$$g(x) = \frac{1}{1 + \mathbf{e}^{(-5\,x)}}$$



$$a \rightarrow \Big[(10\,\mathbf{e}^{(-5/2)}\,(\mathbf{e}^{a})^{10}\,a\,\mathbf{e}^{(15/2)} + 3\,\mathbf{e}^{(-5/2)}\,\mathbf{e}^{(15/2)} - 10\,\mathbf{e}^{(-5/2)}\,(\mathbf{e}^{a})^{10}\,a\,\mathbf{e}^{(5/2)}$$
$$+ 7\,\mathbf{e}^{(-5/2)}\,(\mathbf{e}^{a})^{10}\,\mathbf{e}^{(5/2)} + 3\,\mathbf{e}^{(-5/2)}\,(\mathbf{e}^{a})^{10}\,\mathbf{e}^{(15/2)} + 7\,\mathbf{e}^{(5/2)}\,\mathbf{e}^{(-5/2)} + 20\,\mathbf{e}^{(-5/2)}\,(\mathbf{e}^{a})^{5}\,\mathbf{e}^{5}$$
$$+ 2\,\mathbf{e}^{(-5/2)}\,(\mathbf{e}^{a})^{5} + 10\,\mathbf{e}^{(-5/2)}\,a\,\mathbf{e}^{(5/2)} - 10\,\mathbf{e}^{(-5/2)}\,a\,\mathbf{e}^{(15/2)} - 2\,\mathbf{e}^{(-5/2)}\,(\mathbf{e}^{a})^{5}\,\mathbf{e}^{10}) \Big/ ($$
$$10\,\mathbf{e}^{5}\,(\mathbf{e}^{a})^{10} - 10\,\mathbf{e}^{5} - 10\,(\mathbf{e}^{a})^{10} + 10),\ \frac{\mathbf{e}^{(-5/2)}\,(\mathbf{e}^{a})^{5} - \mathbf{e}^{(-5/2)}\,(\mathbf{e}^{a})^{5}\,\mathbf{e}^{5} + 5\,\mathbf{e}^{(-5/2)}\,(\mathbf{e}^{a})^{10}\,\mathbf{e}^{(5/2)}}{5\,(\mathbf{e}^{a})^{10} - 5} \Big]$$

```
> dsolve( { D(g)(x)=5*g(x)*(1-g(x)), g(0)=1/2 } );
  untersuche( unapply(subs(%,diff(g(x),x)),x),
```

```
[-1.2..1.2,-0.2..1.4] );
```

$$g(x)=\frac{1}{1+\mathbf{e}^{(-5\,x)}}$$



$$\begin{aligned}
a \rightarrow [(&3\,\mathbf{e}^{10}+34\,\mathbf{e}^{5}\,(\mathbf{e}^{a})^{10}-3\,(\mathbf{e}^{a})^{10}+7\,\mathbf{e}^{5}+10\,(\mathbf{e}^{a})^{20}\,a\,\mathbf{e}^{15}-10\,a\,\mathbf{e}^{10}+10\,a\,\mathbf{e}^{5}\\
&+26\,\mathbf{e}^{10}\,(\mathbf{e}^{a})^{10}-34\,\mathbf{e}^{5}\,(\mathbf{e}^{a})^{20}-7\,\mathbf{e}^{5}\,(\mathbf{e}^{a})^{30}-26\,\mathbf{e}^{10}\,(\mathbf{e}^{a})^{20}+4\,(\mathbf{e}^{a})^{25}\,\mathbf{e}^{(25/2)}-3\,(\mathbf{e}^{a})^{30}\,\mathbf{e}^{10}\\
&-4\,(\mathbf{e}^{a})^{25}\,\mathbf{e}^{(5/2)}+3\,(\mathbf{e}^{a})^{20}+60\,(\mathbf{e}^{a})^{20}\,a\,\mathbf{e}^{10}-60\,(\mathbf{e}^{a})^{20}\,a\,\mathbf{e}^{5}-80\,(\mathbf{e}^{a})^{15}\,a\,\mathbf{e}^{(5/2)}\\
&+80\,(\mathbf{e}^{a})^{15}\,a\,\mathbf{e}^{(25/2)}+10\,(\mathbf{e}^{a})^{10}\,a\,\mathbf{e}^{15}-60\,(\mathbf{e}^{a})^{10}\,a\,\mathbf{e}^{5}+60\,(\mathbf{e}^{a})^{10}\,a\,\mathbf{e}^{10}+10\,(\mathbf{e}^{a})^{30}\,a\,\mathbf{e}^{5}\\
&-40\,(\mathbf{e}^{a})^{25}\,\mathbf{e}^{(15/2)}+7\,(\mathbf{e}^{a})^{20}\,\mathbf{e}^{15}+4\,\mathbf{e}^{(5/2)}\,(\mathbf{e}^{a})^{5}-7\,(\mathbf{e}^{a})^{10}\,\mathbf{e}^{15}+40\,\mathbf{e}^{(15/2)}\,(\mathbf{e}^{a})^{5}\\
&-4\,(\mathbf{e}^{a})^{5}\,\mathbf{e}^{(25/2)}-10\,(\mathbf{e}^{a})^{10}\,a-10\,(\mathbf{e}^{a})^{20}\,a-10\,(\mathbf{e}^{a})^{30}\,a\,\mathbf{e}^{10})\Big/(10\,\mathbf{e}^{10}\,(\mathbf{e}^{a})^{10}\,\mathbf{e}^{5}\\
&+70\,(\mathbf{e}^{5})^{2}\,(\mathbf{e}^{a})^{20}+80\,\mathbf{e}^{5}\,(\mathbf{e}^{a})^{15}\,\mathbf{e}^{(5/2)}-10\,(\mathbf{e}^{5})^{2}\,(\mathbf{e}^{a})^{30}+80\,\mathbf{e}^{(15/2)}\,(\mathbf{e}^{a})^{15}\,\mathbf{e}^{5}-60\,\mathbf{e}^{5}\,(\mathbf{e}^{a})^{10}
\end{aligned}$$

$$+10\,\mathbf{e}^5\,(\mathbf{e}^a)^{20}\,\mathbf{e}^{10}-60\,\mathbf{e}^5\,(\mathbf{e}^a)^{20}-10\,(\mathbf{e}^5)^2+70\,(\mathbf{e}^5)^2\,(\mathbf{e}^a)^{10}-10\,\mathbf{e}^{10}\,(\mathbf{e}^a)^{10}-80\,\mathbf{e}^{(5/2)}\,(\mathbf{e}^a)^{15}$$

$$+10\,\mathbf{e}^5\,(\mathbf{e}^a)^{30}-80\,\mathbf{e}^{(15/2)}\,(\mathbf{e}^a)^{15}-10\,(\mathbf{e}^a)^{10}-10\,\mathbf{e}^{10}\,(\mathbf{e}^a)^{20}-10\,(\mathbf{e}^a)^{20}+10\,\mathbf{e}^5),\ ($$

$$10\,(\mathbf{e}^{(5/2)})^2\,(\mathbf{e}^a)^{10}-\mathbf{e}^{(5/2)}\,(\mathbf{e}^a)^{25}\,\mathbf{e}^5+8\,\mathbf{e}^{(5/2)}\,(\mathbf{e}^a)^{15}+12\,\mathbf{e}^5\,(\mathbf{e}^a)^{15}\,\mathbf{e}^{(5/2)}+10\,(\mathbf{e}^{(5/2)})^2\,(\mathbf{e}^a)^{20}$$

$$+\mathbf{e}^{(5/2)}\,(\mathbf{e}^a)^5-\mathbf{e}^{(5/2)}\,(\mathbf{e}^a)^5\,\mathbf{e}^5+(\mathbf{e}^a)^{25}\,\mathbf{e}^{(5/2)})\Big/(\mathbf{e}^{10}\,(\mathbf{e}^a)^{10}+7\,\mathbf{e}^5\,(\mathbf{e}^a)^{20}+8\,\mathbf{e}^{(5/2)}\,(\mathbf{e}^a)^{15}$$

$$-\mathbf{e}^5\,(\mathbf{e}^a)^{30}+8\,\mathbf{e}^{(15/2)}\,(\mathbf{e}^a)^{15}+(\mathbf{e}^a)^{10}+\mathbf{e}^{10}\,(\mathbf{e}^a)^{20}+(\mathbf{e}^a)^{20}-\mathbf{e}^5+7\,\mathbf{e}^5\,(\mathbf{e}^a)^{10})]$$

```
> H:=untersuche( x->sqrt(3-x^2), [-1.2..1.2,0..2] );
```

Achtung! Diese Funktion ist besonders eklig, weil sie für x-Werte größer als $\sqrt{3}$ nicht definiert (bzw. nicht reell) ist. Da wir Sekanten an der angegebenen Stelle plus/minus $\frac{1}{2}$ abgreifen, muss das entsprechend verlängerte x-Intervall im Definitionsbereich liegen, damit alles funktioniert. Andererseits dürfen wir auch mit 'piecewise' oder 'abs' nicht allzu verschwenderisch sein, da sonst Maples 'solve' Schwierigkeiten bekommt.



$$H := a \to \left[\left(-6\sqrt{11-4a^2-4a}+4\sqrt{11-4a^2-4a}\,a^2-2\sqrt{11-4a^2-4a}\,a+6\sqrt{11-4a^2+4a}-4\sqrt{11-4a^2+4a}\,a^2-2\sqrt{11-4a^2+4a}\,a\right) \Big/ \left(-\sqrt{11-4a^2-4a}-\sqrt{11-4a^2+4a}-2\sqrt{11-4a^2+4a}\,a+2\sqrt{11-4a^2-4a}\,a\right), \left(11\sqrt{11-4a^2+4a}\sqrt{11-4a^2-4a}-4\sqrt{11-4a^2+4a}\sqrt{11-4a^2-4a}\,a^2-143+96a^2-16a^4\right) \Big/ \left(-4\sqrt{11-4a^2+4a}\,a-2\sqrt{11-4a^2+4a}+4\sqrt{11-4a^2-4a}\,a-2\sqrt{11-4a^2-4a}\right)\right]$$

Auflösen?

```
> H:=untersuche( x->1/x, [-5.001..5,-5..5], scaling=constrained
  );
```



$$H := a \to \left[\frac{1}{4}\,\frac{4a^2+1}{a}, \frac{1}{a}\right]$$

Hier ist Auflösen nach x aufschlussreicher:

```
> _EnvExplicit:=true:
```

```
  H(a):
  solve( { x=%[1], y=%[2] }, {a,x} );
```

$$\{a = \frac{1}{y}, x = \frac{4 + y^2}{4\, y}\}$$

Die Hüllkurve ist wiederum eine Hyperbel.

```
> x=subs( %, x ):
  % * denom(rhs(%));
```

$$4\, x\, y = 4 + y^2$$

```
>
```